ICFF-V Symposium Paper, August 17 - 21, 1997
National Institue of Standards and Technology
Gaithersburg, MD 20899

CONF-970877--1

THE DISLOCATION FREE ZONE MODEL OF FRACTURE BY SYMBOLIC PROGRAMMING[1]

Shih-Jung Chang
Research Reactors Division
Oak Ridge National Laboratory
Oak Ridge, TN 37831-6399 USA

Abstract

The dislocation free zone (DFZ) model of fracture was developed by Chang and Ohr (1,2) after a series of experimental observations on the crack tip dislocation structures that invariably showed the existence of the dislocation free region (3,4). The DFZ model is a modified BCS crack model that is supplemented with the Rice-Thomson crack tip dislocation emission mechanism. This dislocation emission mechanism imposes a finite energy barrier to the crack tip for emitting dislocations into the plastic zone, in contrast to a zero energy barrier for the BCS model. This finite energy barrier results in the formation of the DFZ and a stress-singular crack tip region. This resistance was expressed in terms of a dislocation emission toughness K_e as a material constant. Because of the emission toughness K_e, the crack tip has the choice either to emit dislocation or to fracture in brittle mode. The model, therefore, was first used by them to explain the fundamental phenomenon of brittle versus ductile fracture. Brittle fracture occurs if $K_{Ic} < K_e$, that is, the crack tip breaks before the dislocation can be emitted. Ductile fracture is possible if $K_{Ic} > K_e$ so that dislocation will be generated before brittle fracture toughness K_{Ic} is reached.

The distribution function for the dislocations was solved from the dislocation pile-up equation. It was expressed in terms of the complete elliptic integrals. Although the analytical nature of the model is clear and precise, but the numerical values of the model may not always be obtained readily. It was attempted to simplify and approximately represent the model by elementary mathematical functions (Louat, Second International Conference on the Fundamentals of Fracture at ORNL, 1985). In this paper the distribution function is written in terms of a symbolic programming language MAPLE. The analytical and numerical manipulations can be made easily. An earlier version of this paper was presented in Micromechanics of Advanced Materials, Symposium for J.C.M. Li 70th Birthday, edited by Peter K. Liaw, TMS 1995. An improvement of the program that accounts for the technique of calculating the elliptic integral of the third kind in different regions of the model is presented here. It is seen that the distribution function shown in this paper is a new form and it has a unique expression representing the distribution of dislocations in different sectors of the crack plane.

The model has been extended to the inclined dislocation free zone cracks (5-7). The inclined DFZ model was then applied to model the threshold region of the fatigue crack growth curve (8). An extensive review on the subject was made by Thomson (9). Lin and Thomson (10)made important progress on the physical implications of the crack tip dislocation emission mechanisms. Li, Dai, Chu and Lee made significant contribution in formulating the discrete dislocation free zone model and the dynamic dislocation free zone model (11-15). Recently, Hirsch, Roberts and Samuels have established the physical basis for the brittle-ductile transition (BDT) temperature of fracture by using the dynamic dislocation free zone model (16). The point of maximum dislocation density, defined as the trigger point, was regarded as a point to initiate ductile fracture by A.T. Yokobori, Jr (17). They made use of the dislocation free zone model to discuss the initiation of ductile fracture. This concept was also mentioned at the end of reference (2) where the significance of the point of maximum dislocation density had been suggested as a point of ductile fracture.

Distribution Function for the Dislocations

The dislocation pile-up integral equation for the model and the condition for the existence of the solution to the integral equation are shown in this section. The distribution function that is the solution to the integral equation is also shown. The existence of the dislocation free zone gives a non-zero stress intensity factor. This stress intensity factor is expressed in terms of the complete elliptic integral of the first kind. The dislocation emission toughness K_e is determined by the Rice-Thomson theory that provides the energy barrier required for a dislocation to be emitted from the crack tip. These equations and the solutions give a complete mathematical description of the dislocation free zone model of fracture.

[1]Based on work performed at Oak Ridge National Laboratory, managed by Lockheed Martin Energy Systems. Inc., for the U.S. Department of Energy under contract DE-AC05-84OR21400. Accordingly, the U.S. government retains a nonexclusive, royal-free license to publish or reproduce the published form of this contribution, or allow others to do so, for U.S. government purposes.

DISTRIBUTION OF THIS DOCUMENT IS UNLIMITED

MASTER

DISCLAIMER

Portions of this document may be illegible in electronic image products. Images are produced from the best available original document.

DISCLAIMER

This report was prepared as an account of work sponsored by an agency of the United States Government. Neither the United States Government nor any agency thereof, nor any of their employees, make any warranty, express or implied, or assumes any legal liability or responsibility for the accuracy, completeness, or usefulness of any information, apparatus, product, or process disclosed, or represents that its use would not infringe privately owned rights. Reference herein to any specific commercial product, process, or service by trade name, trademark, manufacturer, or otherwise does not necessarily constitute or imply its endorsement, recommendation, or favoring by the United States Government or any agency thereof. The views and opinions of authors expressed herein do not necessarily state or reflect those of the United States Government or any agency thereof.

To describe the model, the crack is located along the x-axis in the region $-c \le x \le c$. The plastic zones are $-a \le x \le -e$ and $e \le x \le a$ and the dislocation free zones are $-e \le x \le -c$ and $c \le x \le e$. For a uniformly applied stress σ at infinity, the pile-up integral equation is

$$\int_{-a}^{-e} + \int_{-c}^{c} + \int_{e}^{a} \frac{Af(z)dz}{x - z} = -G(x)$$

$$G(x) = \sigma \qquad \text{for} \quad -c < x < c$$

$$G(x) = \sigma - \sigma_f \quad \text{for} \quad -a < x < -e \text{ and } e < x < a$$

In the above equation, σ_f is the friction stress and $A = \mu b / 2\pi\kappa$. The symbol κ is equal to $1 - \nu$ for Mode I and Mode II, where ν is the Poisson's ratio, and κ is equal to 1 for Mode III. Furthermore, b is the Burger's vector and μ is the shear modulus.

The distribution was solved from the integral equation in terms of the complete elliptic integrals. In this paper, the earlier distribution function is rewritten and the discontinuous sectors are patched together and expressed everywhere along the x-axis between $-a$ and a by a single distribution function with the following form

$$f(x) = \frac{\sigma_f}{A\pi^2} \frac{2\sqrt{k^2 - y^2}}{y\sqrt{1 - y^2}} [(1 - y^2)\Pi(y^2, k^2) - F(k^2)]$$

where F and Π are complete elliptic integrals of the first kind and the third kind, respectively. The above solution was under the condition of finite stress in the plastic zone, defined as the DFZ condition, of the form

$$\frac{\sigma}{\sigma_f} = \frac{2\sqrt{\alpha^2 - k^2}\sqrt{1 - \alpha^2}}{\pi\alpha} \Pi(\alpha^2, k^2)$$

The stress intensity factor at the crack tip was also obtained as

$$\frac{K}{\sigma_f\sqrt{\pi c}} = \frac{2\sqrt{\alpha^2 - k^2}}{\pi\alpha} F(k^2)$$

In the above equations,

$$\alpha^2 = \frac{a^2 - e^2}{a^2 - c^2}$$

$$y^2 = \frac{(a^2 - x^2)\, c^2}{(a^2 - c^2)\, x^2}$$

and

$$k^2 = \alpha^2 \frac{c^2}{e^2} .$$

The complete elliptic integral of the first kind is defined as

$$F(k^2) = \int_0^{\pi/2} \frac{dt}{\sqrt{1 - k^2 \sin^2(t)}}$$

and the complete elliptic integral of the third kind is defined as

$$\Pi(y^2, k^2) = \int_0^{\pi/2} \frac{dt}{(1 - y^2 \sin^2(t)) \sqrt{1 - k^2 \sin^2(t)}}$$

Dislocation Free Zone Model By MAPLE

By using the symbolic programming language MAPLE, it is not difficult to analytically and numerically manipulate the above solution and conditions. The numerical values need to be evaluated from the model may be obtained once the elliptic integrals are evaluated. By using MAPLE, the calculation is straight forward. Also, simple integration and differentiation may be carried out rather easily by MAPLE commands INT and DIFF. The distribution function shown above can be integrated by MAPLE integration command INT to verify numerically that the distribution function is indeed the solution of the pile-up integral equation. The point of maximum dislocation density, the triger point for ductile fracture, can be obtained by DIFF command that is applied to the distribution function to locate the maximum value of the distribution function. The value of the crack openning displacement (COD) can be obtained by using INT command to the distribution function. The integration gives the total number of dislocations and, after multiplied by the Burgers vector, the value of the COD. The MAPLE procedure for the dislocation free zone model is shown in the following.

To execute the program use the command "read *maplefile*;". It will generate plot as shown in the following figure. In the figure, the distribution function is expressed in terms of $A\pi^2 f(x/c) / \sigma_f$ with a/c=1.6 and e/c=1.06.



Fig. 1. Dislocation density function.

```
#######
#######
#
# DISLOCATION FREE ZONE MODEL OF FRACTURE
# BY MAPLE
#
# by Shih-Jung Chang
# January 20, 1997
#
#######
#
# definition of y-array
#
y:=array(1..40):
#
# ff(kk) = elliptic integral - 1st kind
#
# kk is assumed to be less than one
#
ff :=proc(kk) int(1/(1-kk*(sin(t))^2)^(1/2),t=0..Pi/2) end:
#
# function ppf will be integrated in procedure pp and either
# yy or kk is assumed to be less than one.
#
ppf:=proc(x,yy,kk)
 1/((1-yy*(sin(x))^2)*(1-kk*(sin(x))^2)^(1/2))
 end:
#
# pp(yy,kk) = elliptic integral - 3rd kind -
#
# case 1. 0<yy<kk<1
#
# use this procedure and find pp(yy,kk).
#
# case 2. 0<kk<1<yy
#
# use an identity that can be derived from Byrd and Friedman,
# Equs.(414.01) and (415.01), i.e.,
#
#     pp(yy,kk)=-pp(kk/yy,kk)+ff(kk)
#
# where 0<kk<1<yy.
# In the above equation, calculate pp(kk/yy,kk),
# where kk/yy<1, by using this procedure.
#
# case 3. 0<kk<yy<1
#
# use this procedure and find pp(yy,kk).
# this case is used for dfz condition.
# also possible, use Byrd and friedman (117.02) identity
#
# pp(yy,kk)=-pp(kk/yy)+ff(kk)+Pi/2*sqrt(yy/((1-yy)*(yy-kk)))
#
pp :=proc(yy,kk)  int(ppf(t,yy,kk),t=0..Pi/2)  end:
#
# DFZ condition
#
dfz:=proc(a,e)
local kk,alfa,ac,ec;
ac:=a*a; ec:=e*e;
alfa:=(ac-ec)/(ac-1);  # 0<kk<alfa<1
kk:=alfa/ec;
 2/Pi*((1-1/ec)*(1-alfa))^(1/2)*pp(alfa,kk)
# for DFZ cond: above expression = sig/sigf
end:
#
# stress intensity factor
#
sif:=proc(a,e)
local kk,ac,ec;
ac:=a*a; ec:=e*e;
kk:=(ac-ec)/((ac-1)*ec);  # 0<kk<1
 2/Pi*(1-1/ec)^(1/2)*ff(kk)
#for stress intensity: above expression = K/(sigf*(Pi*c)^(1/2))
end:
#
# f(xp) = distribution function at xp
# with DFZ = e and PZ (plastic zone) = a
#
f := proc(xp,a,e)
local xx,yy,kk,alfa,ac,ec,xc,gg;
ac:=a*a; ec:=e*e; xc:=xp*xp;
yy:=(ac-xc)/((ac-1)*xc);
alfa:=(ac-ec)/(ac-1);
kk:=alfa/ec;
xx:=kk/yy;
#
gg:=-2*(kk-yy)^(1/2)*xp/(((ac-xc)/(ac-1))
    ^(1/2)*(1-yy)^(1/2));
#
 if  yy<1 then
  p3:=pp(yy,kk):
 elif yy>1 then
  p3:=-pp(xx,kk)+ff(kk):
 fi:
```

```
# for e<xp<a that implies 0<yy<kk<1 use
 gg*((1-yy)*p3-ff(kk))
# for 0<xp<c that implies yy>1>kk>0 use
# -gg*((1-yy)*pp(xx,kk)+yy*ff(kk))
# that has the same expression
end:
#
# differentiating the function f with respect to xx, use
#
# diff(f(xx,a,e),xx):
#
# evaluating the value of f, use
#a:=1.6: e:=1.06:
#for i from 1 to 10 do
#xx:=0.1+(i-5)*0.2;
#xx:=0.4;
#evalf(f(xx,a,e),3) od;
#
# integrating the function f from e to a to obtain the total
# number of dislocations from e to a, use
#
# int(f(xx,a,e),xx=e..a);
#
g:=proc(x)
local a,e;
a:=1.6; e:=1.06;
f(x,a,e)
end:
#
#plot(g,1.0..1.6);
#interface(plotdevice=ps,plotoutput=dislxps);
#plotting the distribution function of a complete range of x,
#use
plot(g,-1.7..1.7,-5..5,labels=['crack_plane','distribution'],
   title='dislocation_density_distribution');
#
################
################
```

References

1. S.-J. Chang and S. M. Ohr, Dislocation Modelling of Physical Systems, M. F. Ashby, R. Bullough, C. S. Hartley, and J. P. Hirth, Eds.(Pergamon Press, New York, 1981), 23–27

2. S.-J. Chang and S. M. Ohr, J. of Applied Physics, 52(1981), 7174–7181

3. S. M. Ohr, J. A. Horton, and S.-J. Chang, Defects, Fracture, and Fatigue, G. Sih, Ed. (Martinus Nijoff Publisgers, The Hague, The Netherlands, 1983), 3–15

4. S. M. Ohr, Material Sciences and Engineering, 72(1985), 1–35

5. S.-J. Chang and S. M. Ohr, J. of Applied Physics, 55 (1984), 3505-3513

6. S. M. Ohr, S.-J. Chang, and R. Thomson, J. of Applied Physics, 57(1985), 1839

7. S.-J. Chang and T. Mura, International J. Engineering Science, 25(1987), 561–576

8. S.-J. Chang and T. Mura, Advances in Fracture Research, ICF7, K. Salama, Ed., 1989, 1087–1093

9. R. Thomson, Solid State Physics, H. Ehrenreich and D. Turnbull, Eds., vol 39, 1986, 1–129

10. I.-H. Lin and R. Thomson, Acta Metallurgica, Overview 47, 34(1986), 187–206

11. J. C. M. Li, Dislocation Modelling of Physical Systems, M. F. Ashby, R. Bullough, C. S. Hartley, and J. P. Hirth, Eds. (Pergamon Press, New York, 1981), 45–55

12. J. C. M Li, Dislocations in Solids, Yamada Conference IX, H. Susuki, T. Ninomiya, K. Sumino, and T. Takeuchi, Eds. (University of Tokyo, Tokyo, 1985), 617–620

13. Shu-Ho Dai and J. C. M. Li, Scripta Metallurgica, 16(1982), 183–188

14. San-Boh Lee, Engineering Fracture Mechanics, 22(1985), 429

15. S. N. G. Chu, J. of Applied Physics, 53(1982)8678

16. P. B. Hirsch, S. G Roberts, J. Samuels, Proc. R. Soc. London, A421(1989), 25–53

17. A. T. Yokobori, Jr., T. Iwadate, and T. Isogai, Fracture Mechanics: 24th Volume, J. D. Landes, et al., Eds. (ASTM STP 1207, 1994), 464–477